```
Off[Solve::ifun]
```

The constant of Carlen & Loss

```
f[R_] := R^2 A + R^-d B

Solve[f'[R] == 0, R][[1]]
FullSimplify[PowerExpand[f[R] /. %]]
% /. A → K/λ;

FullSimplify[PowerExpand[%^((2+d)/d) /. B → M^2/ω]]
```

$$\left\{R \to 2^{-\frac{1}{2+d}} \left(\frac{B\, d}{A}\right)^{\frac{1}{2+d}}\right\}$$

$$4^{-\frac{1}{2+d}} A^{\frac{d}{2+d}} B^{\frac{2}{2+d}} d^{-\frac{d}{2+d}} (2+d)$$

$$\frac{4^{-1/d} (2+d)^{1+\frac{2}{d}} K M^{4/d} \omega^{-2/d}}{d \lambda}$$

Numerical computation of the eigenvalue

```
ϵ = 10^-6;

LambdaNum[d_] :=
 s^2 /. FindRoot[u'[s] /. NDSolve[{u''[r] + (d - 1)/r u'[r] + u[r] == 0, u[ϵ] == 1,
      u'[ϵ] == 0}, {u, u'}, {r, ϵ, 10}], {s, 3, 6}]

P0 = Plot[LambdaNum[d], {d, 1, 8}, PlotRange → {{0, 8}, {0, 60}}, PlotStyle → Black]
```



```
Show[P0,
 Plot[x^2 /. FindRoot[-1/2 x (BesselJ[-1 + 1/2 (-2 + d), x] - BesselJ[1 + 1/2 (-2 + d), x]) +
     1/2 (-2 + d) BesselJ[1/2 (-2 + d), x], {x, 2 + d/4, 6 + d/4}], {d, 1, 8},
  PlotRange → {{0, 8}, {0, 60}}, PlotStyle → {Black, Thick, Dotted}]]
```




```
Lambda[d_] := x^2 /. FindRoot[BesselJ[d/2 , x], {x, 2 + d/4, 6 + d/4}]
```

```
Show[P0, Plot[Lambda[d], {d, 1, 8},
  PlotRange → {{0, 8}, {0, 60}}, PlotStyle → {Black, Thick, Dotted}]]
```



The optimal constant in Nash’s inequality

```
CNash[d_] := ((d+2)/d Gamma[d/2 + 2]^(2/d)) / (π Lambda[d])
```

The constant deduced from the Sobolev inequality

$$\text{CSobolev}[d\_] := \frac{1}{d\,(d-2)\,\pi}\left(\frac{\text{Gamma}[d]}{\text{Gamma}\left[\frac{d}{2}\right]}\right)^{\frac{2}{d}}$$

The constant deduced from the logarithmic Sobolev inequality

$$\text{CLogSobolev}[d\_] := \frac{2}{\pi\, d\, e}$$

The constant of Stein (Nash)

$$f[R\_] := R^2 A + R^{-d} B$$

```
Solve[f'[R] == 0, R][[1]]
FullSimplify[PowerExpand[f[R] /. %]]
```

$$\text{FullSimplify}\left[\text{PowerExpand}\left[\%^{\frac{2+d}{d}} \;/.\; B \to \frac{\omega M^2}{(2\pi)^d}\right]\right]$$

$$\text{FullSimplify}\left[\text{PowerExpand}\left[\% \;/.\; \omega \to \frac{2\pi^{\frac{d}{2}}}{d\,\text{Gamma}\left[\frac{d}{2}\right]}\right]\right]$$

$$\left\{R \to 2^{-\frac{1}{2+d}}\left(\frac{B\,d}{A}\right)^{\frac{1}{2+d}}\right\}$$

$$4^{-\frac{1}{2+d}} A^{\frac{d}{2+d}} B^{\frac{2}{2+d}} d^{-\frac{d}{2+d}} (2+d)$$

$$\frac{4^{-1-\frac{1}{d}} A\,(2+d)^{1+\frac{2}{d}} M^{4/d}\,\omega^{2/d}}{d\,\pi^2}$$

$$\frac{A\left(\frac{d}{2+d}\right)^{-\frac{2+d}{d}} M^{4/d}\,\text{Gamma}\left[\frac{d}{2}\right]^{-2/d}}{4\pi}$$

$$\text{CStein}[d\_] := \frac{\left(\frac{d}{2+d}\right)^{-\frac{2+d}{d}}\text{Gamma}\left[\frac{d}{2}\right]^{-2/d}}{4\pi}$$

Comparison of the estimates

```
Show[Plot[CStein[d], {d, 1, 8},
   PlotStyle → {Dotted, Thick, Black}, PlotRange → {{0, 8}, All}],
  Plot[CLogSobolev[d], {d, 1, 8}, PlotRange → {{0, 8}, All},
   PlotStyle → {Dashed, Thick, Black}], Plot[CSobolev[d], {d, 2, 8},
   PlotStyle → {Dashed, Thick, Black}, PlotRange → {{0, 8}, {0, 0.8}}],
  Plot[CNash[d], {d, 1, 8}, PlotStyle → {Thick, Black}],
  PlotRange → {{0, 8}, {0, 0.4}}, AxesOrigin → {0, 0}]
```